Relatório - Lista D

Grupo 17
Bruno Caixeta Piazza - 11260551, Gustavo Nunes Ribeiro - 11262212,
Murilo Camargo Marchioni - 11260717, Vinícius Maia Neto - 11261785

28 de setembro de 2021

1 Exercício 1: simulação de um reservatório

Primeiramente, será simulado o sistema de um único reservatório. Para tanto, toma-se como base o código presente no enunciado da lista, apenas adicionando-se o trecho referente ao sistema não linear. A simulação retorna a variação do nível de água em relação ao equilíbrio, $x(t) = h(t) - \bar{h}$, sendo $\bar{h}$ o nível de água no reservatório no ponto de equilíbrio. Assim, obtém-se a equação não linearizada

$$\dot{x} = \frac{1}{S}\left(Q_e - \sqrt{\frac{\rho g(x+\bar{h})}{R}}\right)$$

e a linearizada

$$\dot{x} = -\frac{1}{2S}\sqrt{\frac{\rho g}{R\bar{h}}}x + \frac{1}{S}u$$

A função $u(t)$ é dada pela expressão

$$u(t) = Q_e(t) - \bar{Q}_e$$

A primeira simulação feita foi aquela requerida no exercício 1, com uma vazão de entrada nula e o reservatório partindo do nível $h = 2\ m$. Nessa situação, tem-se que $u(t) = -\bar{Q}_e = 0$. Escolheu-se um tempo de simulação de 40000 s, tempo suficiente para o reservatório praticamente esvaziar-se.

A figura 1 ilustra os resultados obtidos.

Figura 1: Evolução temporal da variação de nível



Fonte: autoria própria.

Nota-se que as curvas mantêm-se próximas nos estágios iniciais. Com a evolução da simulação, porém, as diferenças tornam-se visíveis. O modelo linear decai mais rapidamente que o não linear, como se percebe das curvas.

2 Exercício 2: simulação de um capacitor

Em seguida, simulou-se o circuito descrito pela figura 2.

Figura 2: Circuito a ser simulado



Fonte: enunciado da Lista D.

A equação do circuito é dada por

$$\dot{x} = Ax + Bu \text{ em que } \begin{cases} x = e(t) \\ u = e_i \end{cases} \text{ e } \begin{cases} A = -\dfrac{1}{RC} \\ B = \dfrac{1}{RC} \end{cases}$$

Nota-se que a equação é da mesma forma da equação linearizada do modelo de um reservatório 1. Os parâmetros utilizados para simulação seguem.

- $e_i = 12\ V$
- $C = 100\ \mu F$
- $R = 100\ \Omega$
- $e_0 = 0\ V$
- $t_f = 0.1\ s$

Assim, obtiveram-se os resultado expressos na figura 3.

Figura 3: Evolução temporal da tensão no capacitor



Fonte: autoria própria.

Os resultados são, qualitativamente, análogos ao sistema linearizado do reservatório: a tensão tende a um valor (a tensão da bateria) conforme o tempo tende ao infinito. Tal fato corrobora a analogia dos sistemas hidráulicos e elétricos. Conforme se conhece da solução analítica, trata-se de uma função exponencial. O gráfico se assemelha ao que indica o nível de água no reservatório em função do tempo, conforme feito nas listas B e C.

3 Exercício 3: simulação de dois reservatórios

A etapa seguinte consistiu na implementação em *Scilab* do modelo linear desenvolvido na Lista C para o sistema de dois reservatórios. Retomando-se o trabalho anterior, na forma matricial $\dot{\boldsymbol{x}} = \boldsymbol{A}\boldsymbol{x} + \boldsymbol{B}u$, isto é,

$$\dot{\boldsymbol{x}} = \boldsymbol{A}\boldsymbol{x} + \boldsymbol{B}u \iff$$
$$\begin{bmatrix} \dot{x_1} \\ \dot{x_2} \end{bmatrix} = \frac{\rho g}{2\bar{Q}_e} \begin{bmatrix} -\frac{1}{S_1 R_a} & \frac{1}{S_1 R_a} \\ \frac{1}{S_2 R_a} & \frac{(R_a + R_s)}{S_2 R_a R_s} \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix} + \begin{bmatrix} \frac{1}{S_1} \\ 0 \end{bmatrix} u$$

Seguindo a convenção do enunciado, pode-se escrever uma expressão matricial para indicar as equações algébricas, sendo obtido o vetor de saídas, na forma $\boldsymbol{y} = \boldsymbol{C}\boldsymbol{x} + \boldsymbol{D}u$, ou ainda,

$$\begin{bmatrix} y_1 \\ y_2 \end{bmatrix} = \begin{bmatrix} 1 & 0 \\ 0 & 1 \end{bmatrix} \begin{bmatrix} x_1 \\ x_2 \end{bmatrix}$$

haja vista que $\boldsymbol{D}$ é o vetor nulo, para o caso analisado.

Ante o exposto, de modo análogo ao feito nos exercícios anteriores, implementa-se a função *syslin* para resolver o sistema linearizado. Conforme foi verificado na Lista C, temos que, na situação de equilíbrio, $\bar{h_1} = 4,2\ m$ e $\bar{h_2} = 2,1\ m$. Desse modo, as condições iniciais para a simulação, tomando-se $h_1(0) = h_2(0) = 2\ m$,

$$x_1(0) = 2 - 4,2 = -2,2\ m$$
$$x_2(0) = 2 - 2,1 = -0,1\ m$$

Ademais, para o cálculo de $u(t)$, calculou-se a vazão de equilíbrio com a expressão

$$\bar{Q_e} = \frac{\rho g \bar{h_1}}{R_a}$$

Com os parâmetros de simulação que constam no Apêndice E, obteve-se o gráfico:

Figura 4: Evolução temporal da variação de nível no tanque



Fonte: autoria própria.

Conforme se esperava, conforme o sistema atinge o equilíbrio, as coordenadas $x(t)$ anulam-se. Por fim, deve-se obter o circuito elétrico análogo, usando-se a analogia, $Q \to i$. $p \to V$:

Figura 5: Circuito análogo ao sistema de dois reservatórios



Fonte: autoria própria.

As equações do circuito análogo são:

$$\begin{cases} \left(C_1 D + \dfrac{1}{R_1}\right) V_1 - \dfrac{V_2}{R_1} = i(t) \\ \left(C_2 D + \dfrac{1}{R_2}\right) V_2 - \dfrac{V_1}{R_1} = 0 \end{cases}$$

Anexos

A Funções - 1 reservatório

```
function [xdot] = naolinear(t,x,Qe)
    xdot = (Qe(t) - sqrt((rho * g * (x+ho))/R)) * 1/S;
endfunction

function [Qe] = entrada(t)
    Qe = Qei
endfunction
```

B Simulação - 1 reservatório

```
// Simulação de sistema linear
// Apagando as variáveis anteriores
clear all
// Fechando as janelas de gráficos
clc
close
// Abrindo o diretório
getd(".")

// Definindo parâmetros
S = 10; // [m^2] Área da seção transversal do reservatório
rho = 1000; // [kg/m^3] massa específica da água
g = 10; // [m/s^2] aceleração da gravidade na superfície da Terra
R = 2*10^8; // [Pa/(m^3/s)^2] parâmetro que relaciona pressão e vazão
ho = 2; // [m] nível do reservatório em regime
hi = 0.1; // [m] nível adicional desejado
Qei = 0; // [m^3/s] vazão na entrada

// Definir o sistema linear usando o comando syslin:
A=( -1/(2*S))* sqrt ( rho *g/(R*ho ));
B=1/S;
C=1;
D=0;

tanque = syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema é contínuo no tempo

// Definir a condição inicial:
x0=0; // [m] desvio inicial do nível em relação ao equilíbrio

// Definir o vetor de instantes de tempo:
t=0:10:2500;

// Definir o vetor de entradas:
u = (Qei-sqrt(rho*g*(ho+hi)/R))*ones(t);

// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);


// Simulacao do sistema não linear
xnlinear = ode(x0,t(1),t,list(naolinear,entrada));


scf(1)
// Plotando o resultado linear em azul:
plot(t,y,'b-','LineWidth',2.7)
//Plotano o resultado não linear em vermelho:
plot(t,xnlinear,'r-','LineWidth',2.7);
// Usando a variável do tipo 'lista':
T=list("Resposta do tanque","Variação de nível [m]","Tempo [s]","Modelo Linear","Modelo n
    ão linear");

//Título do gráfico e configuração da fonte:
title(T(1),'fontsize',4);

// Colocando uma legenda na parte inferior direito da figura (parâmetro 2):
legends([T(4),T(5)],[2,5],1,'fontsize',3.5);

// Nomeando os eixos:
```

```
xlabel(T(3),'fontsize',3);
ylabel(T(2),'fontsize',3);

// Colocando uma grade azul no grafico:
xgrid(2)

// Salvando a imagem:
xs2png(1,'caso1.png');
```

C Funções - capacitor

```
function [edot] = tensao(t,e)
    edot = (ei-e)/(R*C)
endfunction
```

D Simulação - capacitor

```
// Simulação de sistema linear
// Apagando as variáveis anteriores
clear
// Fechando as janelas de gráficos
clc
close
// Abrindo o diretório
getd(".")

// Definindo parâmetros
ei = 12.; //Tensão da bateria
C = 10^(-4); //Capacitância
R =100; // Resistância do resistor
e0 = 0.; // Tensão inicial sobre o capacitor

//Vetor de tempo
t = 0:0.005:0.1;

//Chamada do ODE
et = ode(e0,t(1),t,tensao)

scf(1)
// Plotando o resultado em azul:
plot(t,et,'b-','LineWidth',2.7)
// Usando a variável do tipo 'lista':
T=list("Resposta transitória da tensão","Tensão [V]","Tempo [s]");

//Título do gráfico e configuração da fonte:
title(T(1),'fontsize',4);

// Nomeando os eixos:
xlabel(T(3),'fontsize',3);
ylabel(T(2),'fontsize',3);

// Colocando uma grade azul no grafico:
xgrid(2)

// Salvando a imagem:
xs2png(1,'caso1.png');
```

E Simulação - 2 reservatórios

```
// Simulação de sistema linear
// Apagando as variáveis anteriores
clear
// Fechando as janelas de gráficos
clc
close


// Definindo parâmetros
S1 = 10; // [m^2] Área da seção transversal do reservatório 1
S2 = 10; // [m^2] Área da seção transversal do reservatório 2
rho = 1000; // [kg/m^3] massa específica da água
g = 10; // [m/s^2] aceleração da gravidade na superfície da Terra
Ra = 2*10^8; // [Pa/(m^3/s)^2] parâmetro que relaciona pressão e vazão no res. 1
```

```
Rs = 2*10^8; // [Pa/(m^3/s)^2] parâmetro que relaciona pressão e vazão no res. 2
ho = 2; // [m] nível do reservatório em regime
hi = 0.1; // [m] nível adicional desejado
Qei = sqrt ( rho *g*(ho+hi)/Ra); // [m^3/s] vazão na entrada

// Definir o vetor de instantes de tempo:
t=0:10:40000;

// Definicao das matrizes do sistema linearizado
A= rho *g/(2* Qei )*[-1/S1/Ra ,1/S1/Ra ;1/S2/Ra ,-1/S2*(1/Ra +1/Rs)];
B =[1 /S1 ;0];
C =[1 ,0;0 ,1];
D =[0;0];

tanque = syslin('c',A,B,C,D); // o parametro 'c' indica que o sistema é contínuo no tempo

// Definir a condição inicial:
x01 = -2.2; // [m] desvio inicial do nível em relação ao equilíbrio no res.1
x02 = -0.1; // [m] desvio inicial do nível em relação ao equilíbrio no res.1
x0 = [x01;x02];

// V azão de equilíbrio [m^3/s]
Qeo = sqrt ( rho *g*2*(ho+hi)/(Ra+Rs))

// Definir o vetor de entradas:
u = (Qei-Qeo)*ones(t);

// Simulando o sistema usando o comando csim:
[y,x]=csim(u,t,tanque,x0);

scf(1)
// Plotando o resultado linear em azul:
plot(t,x(1,:),'b-','LineWidth',2.7)
//Plotano o resultado não linear em vermelho:
plot(t,x(2,:),'r-','LineWidth',2.7);
// Usando a variável do tipo 'lista':
T=list("Resposta do tanque","Variação de nível [m]","Tempo [s]","Reservatório 1","
    Reservatório 2");

//Título do gráfico e configuração da fonte:
title(T(1),'fontsize',4);

// Colocando uma legenda na parte inferior direito da figura (parâmetro 2):
legends([T(4),T(5)],[2,5],4,'fontsize',3.5);

// Nomeando os eixos:
xlabel(T(3),'fontsize',3);
ylabel(T(2),'fontsize',3);

// Colocando uma grade azul no grafico:
xgrid(2)

// Salvando a imagem:
xs2png(1,'caso1.png');
```